Cranfield
College of Aeronautics Report No. 8719
October 1987

A Fast Riemann Solver with Constant Covolume Applied to the Random Choice Method

E.F. Toro

Department of Aerodynamics
Cranfield Institute of Technology
Cranfield, Bedford MK43 0AL, England

Cranfield
College of Aeronautics Report No. 8719
October 1987

A Fast Riemann Solver with Constant Covolume Applied to the Random Choice Method

E.F. Toro

Department of Aerodynamics
Cranfield Institute of Technology
Cranfield, Bedford MK43 0AL, England

ISBN 0 947767 81 9

£7.50

Abstract

The Riemann problem for the unsteady, one dimensional Euler equations together with the constant-covolume equation of state is solved exactly. The solution is then applied to the Random Choice Method to solve the general initial-boundary value problem for the Euler equations. The iterative procedure to find p*, the pressure between the acoustic waves, involves a single algebraic (non-linear) equation, all other quantities follow directly throughout the x - t plane, except within rarefaction fans, where an extra iterative procedure is required.

1. Introduction

The ideal-gas kinetic theory assumes that molecules occupy a negligible volume and that they do not exert forces on one another. In applications such as in combustion processes, these assumptions are no longer accurate descriptions of the problem. In this paper we incorporate covolume, that is to say, we assume that molecules occupy a finite volume b, so that the volume available for molecular motion is $v - b$. The resulting thermal equation of state is

$$p(v - b) = RT \tag{1}$$

Here p, v, R and T are pressure, volume, the gas constant and absolute temperature respectively, with $v = 1/\rho$; ρ is density.

If one were to assume intermolecular forces as well, then the Van der Waals' equation of state would result. However, we are only interested in eq. (1) where b is constant (with dimensions m^3/kg). Corner [1] reports on experimental results for a good range of solid propellants, where he observed that the covolume b varied very little, i.e. $0.9 \times 10^{-3}\ m^3/kg \leqslant b \leqslant 1.1 \times 10^{-3}\ m^3/kg$. The best values of b lead to errors no greater than 2% and thus we feel there is some justification in using eq. (1) with b = constant, when modelling gas dynamical events associated with solid propellant burning.

The main motivation of the present work is to extend the applicability of the Random Choice Method (RCM) to model gas dynamical events arising from, and coupled with, combustion phenomena. Since RCM uses the exact solution of the Riemann problem, our first task will be to devise an efficient Riemann solver. In Ref. [2] we derived a number of covolume relations and indicated a solution strategy based on the Newton-Raphson Method applied to a 3 x 3 system of algebraic equations. For rarefaction fans we also suggested a similar approach to solve another 3 x 3 system. The resulting Riemann solver was found to be more efficient than that based on the Godunov iteration when applied to the special case b = 0 (ideal gas), but the net gains were limited.

The present Riemann solver is much more efficient; it is an extension of that proposed in Ref.[3] for ideal gases. The two iteration procedures that are present (one the pressure p* between the acoustic waves and the other for the density ρ inside rarefaction fans) involve a single algebraic equation. The Newton-Raphson Method works well in both cases.

The implementation of RCM using the exact Riemann solver is carried out on a non-staggered grid, whereby the solution to the next time level is advanced in a single step. This programming strategy has a number of advantages over the more common staggered grid approach. Simplicity is one of them. Use of irregular/adaptive grids is another. The original idea appears to be due to Colella [4].

The remaining part of this paper is organised as follows: Section 2 defines the Riemann problem and delineates the solution strategy. In section 3 we collect the covolume relations required to solve the problem. In section 4 we solve the Riemann problem. In section 5 we describe the implementation of RCM. In section 6 we solve a shock-tube problem exactly by direct application of the present Riemann solver and approximately via the Random Choice Method. Results are compared and discussed. Finally, in section 7 we draw some conclusions and indicate areas of applications of present results.

2. The Riemann Problem

We consider the Riemann problem for the unsteady one-dimensional Euler equations together with the covolume equation of state (1) with constant b, namely

$$U_t + F(U)_x = 0 \tag{2}$$

$$U(x,t_o) = \begin{cases} U_l \, , & x \leqslant x_o \\ U_r \, , & x \geqslant x_o \end{cases} \tag{3}$$

where $-\infty < x < \infty$, $t > t_o$. Here $U = U(x,t)$ with x and t denoting space and time respectively. In eq. (2) subindices denote partial differentiation, as usual. U and F(U) are vectors of conserved variables and fluxes respectively. These are given by

$$U = \begin{bmatrix} \rho \\ \rho u \\ E \end{bmatrix}, \quad F(U) = \begin{bmatrix} \rho u \\ \rho u^2 + p \\ (E + p)u \end{bmatrix} \tag{4}$$

where u is velocity, e is specific internal energy and E is total energy given by

$$E = \tfrac{1}{2}\rho u^2 + \rho e \tag{5}$$

The initial condition (3) consists of two constant states U_l and U_r.

Note that equation (1) serves as a closure condition for system (2), which has three differential equations and four unknowns. A corresponding caloric equation of state gives an expression for the specific internal energy in eq. (5) in terms of the unknowns of system (2).

The solution of the Riemann problem (1) - (5) for $t > t_o$ can be represented in the half x - t plane as in Fig. 1.



Fig. 1: Solution of Riemann problem with data U_l and U_r

There are three waves present: W_L, W_M and W_R. The middle wave W_M is always a contact discontinuity, the left wave W_L is either a shock or a rarefaction and the right wave W_R is either a shock or a rarefaction. Hence, there are four possible wave patterns. The region star between waves W_L and W_R is characterised by having pressure p^* = constant and velocity u^* = constant with $\rho = \rho_l^*$ between W_L and W_M (star left) and $\rho = \rho_r^*$ between W_M and W_R (star right). In the portion of the half x - t plane to the left of wave W_L the solution is equal to the constant state U_l (data). Similarly $U = U_r$ in the region to the right of wave W_R. The solution U at a time $t > t_o$ inside a rarefaction fan (W_L or W_R) varies smoothly with x.

The principal step of the solution procedure is the determination of the solution in the region star. We call this the star step. A feature of the present Riemann solver is that the star-step consists of a single (non-linear) algebraic equation for the pressure p^*. Other quantities in the region star follow directly. Clearly, the solution for p^* must be found iteratively, since the type of waves W_L and W_R is not known a-priori. This must be determined as part of the solution.

The star-step requires equations connecting U_l (data) to U_l^* and U_r (data) to U_r^*. In each situation one must derive equations for the case in which the connecting wave is a shock or a rarefaction. These equations are manipulated in such a way that the velocities u_l^* and u_r^* are expressed as

$$\begin{aligned} u_l^* &= f_l(p^*, U_l) \\ u_r^* &= f_r(p^*, U_r) \end{aligned} \tag{6}$$

But $u_l^* = u_r^*$ gives a single algebraic non-linear equation for the unknown p^*, i.e.

$$f(p^*, U_l, U_r) \equiv f_l(p^*, U_l) + f_r(p^*, U_r) = 0 \tag{7}$$

A certain amount of work is involved in determining the form of the functions f_l and f_r in equations (6), and thus f in eq. (7).

Once p^* is known from eq. (7) all other quantities in region star follow directly from explicit relations. If both waves W_L and W_R are shocks then the solution of the Riemann problem has been determined everywhere in the half x - t plane. However if a rarefaction fan is present the solution inside it requires another iterative procedure. This is unlike the ideal-gas case, where the solution inside rarefaction fans follows directly from the star step (also iterative). We present an economical way of finding the solution inside rarefaction fans. Instead of solving a 3 x 3 non-linear system (as suggested in Ref. [2]) we solve a single non-linear equation for the density ρ. Other quantities follow directly.

Next, we collect some basic relations for shock and rarefaction waves and derive covolume expressions for the internal energy and the sound speed. These will be later utilised in the star step.

3. Covolume relations

Here we collect some of the covolume equations derived in Ref. 2. There, we showed that the specific internal energy e is given by

$$e = \frac{p(1 - b\rho)}{\rho(\gamma - 1)} \tag{8}$$

and the sound speed c is given by

$$c = \left[\frac{p\gamma}{\rho(1 - b\rho)}\right]^{\frac{1}{2}} \tag{9}$$

Here γ denotes ratio of specific heats, as usual. The derivation of equations across shocks and rarefactions is now dealt with separately.

3.1 Shock relations

Consider the case of a right travelling shock wave of speed S_r. In the steady frame of reference attached to the shock the usual equations for mass momentum and energy apply. In Ref. [2] we formulated the solution of the star step in terms of the pressure p* and two parameters M_l and M_r. In the present paper the solution strategy is different, but expressions for M_r and M_l are still useful. For a right moving wave (shock or rarefaction) M_r is defined as

$$M_r = \frac{p^* - p_r}{u^* - u_r} \tag{10}$$

For a right travelling shock, the steady shock relations give

$$M_r^2 = \frac{\rho_r(p^* - p_r)D_R}{D_R - 1} \tag{11}$$

where $D_R = \rho_r^*/\rho_r$ is the density ratio across the shock wave. Also, the standard Hugoniot relation can be written as

$$e^* - e_r = \tfrac{1}{2}\left(\frac{p_r}{\rho_r}\right)\left[\frac{(P_R + 1)(D_R - 1)}{D_R}\right] \tag{12}$$

where $P_R = p^*/p_r$ is the pressure ratio across the shock. Substitution of e from eq. (8) into eq. (12) gives a relationship between P_R and D_R across the shock i.e.

$$D_R = \frac{(\gamma + 1)P_R + (\gamma - 1)}{(\gamma - 1 + 2b\rho_r)P_R + (\gamma + 1) - 2b\rho_r} \tag{13}$$

which, if used in eq. (11), leads to

$$M_r = \left\{\left[\frac{(\gamma + 1)}{2}\frac{\rho_r\, p_r}{(1 - b\rho_r)}\right]\left[P_R + \frac{(\gamma - 1)}{(\gamma + 1)}\right]\right\}^{\frac{1}{2}} \tag{14}$$

Similarly, for the left travelling wave W_L a parameter M_l can be defined as follows

$$M_l = -\frac{(p^* - p_l)}{(u^* - u_l)} \tag{15}$$

which, after using appropriate relations, becomes

$$M_l = \left\{\left[\frac{(\gamma + 1)}{2}\frac{\rho_l\, p_l}{(1 - b\rho_l)}\right]\left[P_L + \frac{(\nu - 1)}{(\gamma + 1)}\right]\right\}^{\frac{1}{2}} \tag{16}$$

Here $P_L = p^*/p_l$ is the pressure ratio across the left moving shock.

3.2 Rarefaction relations

In order to obtain expressions for M_l and M_r in the case in which waves W_L and W_R are rarefaction waves we need the generalised Riemann invariants and the isentropic relations. For a left rarefaction

$$J_L = u + \frac{2c}{(\gamma - 1)}(1 - b\rho) = \text{constant} \tag{17}$$

and

$$\frac{\rho_l^*}{(1 - b\rho_l^*)} = \frac{\rho_l}{(1 - b\rho_l)} P_L^{1/\gamma} \tag{18}$$

For a right rarefaction we have

$$J_R = u - \frac{2c}{(\gamma - 1)}(1 - b\rho) = \text{constant} \tag{19}$$

and

$$\frac{\rho_r^*}{(1 - b\rho_r^*)} = \frac{\rho_r}{(1 - b\rho_r)} P_R^{1/\gamma} \tag{20}$$

Use of eqs. (17) - (18) gives for M_l

$$M_l = \frac{(\gamma - 1)}{2}\left[\frac{\rho_l\, p_l}{\gamma(1 - b\rho_l)}\right]^{\frac{1}{2}}\left[\frac{1 - P_L}{1 - P_L^{\frac{\gamma-1}{2\gamma}}}\right] \tag{21}$$

and use of eqs. (19) - (20) gives for M_r

$$M_r = \frac{(\gamma - 1)}{2}\left[\frac{\rho_r\, p_r}{\gamma(1 - b\rho_r)}\right]^{\frac{1}{2}}\left[\frac{1- P_R}{1 - P_R^{\frac{\gamma-1}{2\gamma}}}\right] \tag{22}$$

We now return to eq. (6). Note that for a left wave, from definition (15) for M_l we have

$$u^* = u_l + \frac{(p_l - p^*)}{M_l}$$

or

$$u^* = u_l + f_l(p^*,U_l) \tag{23}$$

where

$$f_l = \begin{cases} (1 - P_L)\left[\dfrac{\dfrac{2(1 - b\rho_l)p_l}{(\gamma + 1)\rho_l}}{P_L + \dfrac{\gamma - 1}{\gamma + 1}}\right]^{\frac{1}{2}} & \text{if } P_L \geqslant 1 \text{ (shock)} \quad (24a) \\[2ex] \dfrac{2(1 - b\rho_l)C_l}{(\gamma - 1)}\left[1 - P_L^{\frac{\gamma-1}{2\gamma}}\right] & \text{if } P_L < 1 \text{ (rarefaction)} \quad (24b) \end{cases}$$

Similarly, for a right wave definition (10) gives

$$u^* = u_r - f_r(p^*, U_r) \tag{25}$$

where

$$
f_r = \begin{cases} (1 - P_R)\left[\dfrac{\dfrac{2(1 - b\rho_r)p_r}{(\gamma + 1)\rho_r}}{P_r + \dfrac{\gamma - 1}{\gamma + 1}}\right]^{\frac{1}{2}} & \text{if } P_R \geqslant 1 \text{ (shock)} \qquad (26a) \\[2em] \dfrac{2(1 - b\rho_r)C_r}{(\gamma - 1)}\left[1 - P_R^{\frac{\gamma-1}{2\gamma}}\right] & \text{if } P_R < 1 \text{ (rarefaction)} \qquad (26b) \end{cases}
$$

We have now completely determined the problem for the star-step. From eqs. (23) and (25) the single equation (7) for p^* results, where f_l and f_r are given by eqs. (24) and (26) respectively.

4. Algorithm for the solution of the Riemann problem

Here we use all relations developed in section 3 to implement an efficient algorithm for completely solving the Riemann problem with constant covolume in the half plane x - t.

As pointed out in section 2 the solution procedure consists basically of the star-step and the rarefaction fan step. The principal part of the star step is the solution of an equation for the pressure p^* in region star. The rarefaction fan step consists of finding the complete solution inside a rarefaction fan; its principal step is the solution of a single equation for the density ρ, Both steps contain an iteration. We shall deal with each of them separately.

4.1 The star step

The main part here is the determination of p^* by solving the single non-linear algebraic equation

$$
f(p^*,U_l,U_r) \equiv f_l(p^*,U_l) + f_r(p^*,U_r) + u_l - u_r = 0 \qquad (27)
$$

where f_l and f_r are given by eqs. (24) and (26). We do this by a Newton-Raphson iteration procedure of the form

$$
p^*_{(k)} = p^*_{(k-1)} + \delta_{(k-1)} \qquad (28)
$$

where

$$\delta_{(k)} = -f(p^*_{(k)}, U_l, U_r)/f'_{(k)}$$

Here k denotes the iteration and $\delta_{(k)}$ is an increment at the k-th iteration.

The method requires evaluation of derivatives

$$f'_{(k)} = \frac{d}{dp^*} f(p^*, U_l, U_r)\Bigg|_{p^* = p^*(k)}$$

at the known point $p^* = p^*_{(k)}$ and an initial (guess) value p^*_0. An economical guess value would be $p^*_0 = \frac{1}{2}(p_l + p_r)$, but it could be inaccurate which can increase the number of iterations for convergence. We say that iteration procedure has converged to the solution at iteration k = K if

$$CHA = \frac{|p^*_K - p^*_{(K-1)}|}{p^*_{(K)}} \leqslant TOL \qquad (29)$$

where TOL is a chosen tolerance, e.g. TOL = 10^{-4} is found to give sufficiently accurate solutions.

An accurate (although expensive) guess value p^*_0 can be found if we assume that both acoustic waves W_L and W_R are rarefaction waves, that is in evaluating f_l and f_r in eq. (27) for p^*, eqs. (24b) and (26b) apply. Algebraic manipulations give a closed form solution for p^*_0 as

$$p^*_0 = \left\{\frac{(1 - b\rho_l)C_l + (1 - b\rho_r)C_r + \frac{(\gamma - 1)}{2}(u_l - u_r)}{(1 - b\rho_l)C_l/p_l^{\frac{(\gamma-1)}{2\gamma}} + (1 - b\rho_r)C_r/p_r^{\frac{(\gamma-1)}{2\gamma}}}\right\}^{\frac{2\gamma}{(\gamma-1)}} \qquad (30)$$

Clearly if both W_L and W_R are rarefaction waves, then eq. (30) gives the exact solution for p^*. But even if the assumption leading to eq. (30) is not true the estimate p^*_0 is quite accurate [3] even for cases involving shocks of strength of about 3. The reason for this is that the rarefaction and shock branches of the p - u curve (see Ref. [5]) have 1st and 2nd continuous derivatives at their intersection point. Thus a continuation of, say, a shock branch via the rarefaction branch is a good approximation for data states U_l and U_r that are sufficiently close, in a given sense.



Figure 2: Algorithm for finding p*

If the solution of the Riemann problem is used in a local sense, as applied to the Random Choice Method, then there may well be one or two genuine discontinuities (shocks or contacts) in the flow field at a given time. Thus typically 98% of the local Riemann problems have data with close states and thus p_0^* as given by eq. (30) is very accurate. A single iteration is performed in most, if not all, of these cases.

Fig. 2 illustrates the algorithm for solving eq. (27) for p^*. Once p^* has been found u^* follows directly from any of eqs. (23) or (25). In practice, it is advisable to take a mean value. The determination of ρ_l^* and ρ_r^* (Fig. 1) depends now on the type of waves W_L and W_R. For instance if W_R is a shock wave then ρ_r^* follows directly from eq. (13). If W_L is a shock wave we use the counterpart of eq. (13) to find ρ_l^*. If W_L is a rarefaction then eq. (18) gives ρ_l^*; if W_R is a rarefaction eq. (20) gives ρ_r^*. Thus, the complete solution of the Riemann problem in the region star has been obtained.

A simple but important Riemann problem is that arising at boundaries. The solution has closed form and is given in the next section.

4.2 The Riemann problem at a moving boundary

Consider the right boundary and assume this is given by a piston moving with known speed V_p. If reflections are to be allowed then the following boundary conditions apply

$$\rho_r = \rho_l \ , \quad u_r = -u_l + 2V_p \ , \quad p_r = p_l \tag{31}$$

Here subscript l denotes last grid point inside the computational domain, and subscript r denotes fictitious grid point immediately to the right of the piston.

The Riemann problem with data (31) has solution as depicted in Fig.1 with $u^* = V_p$ and W_L and W_R both of the same type, i.e. they are both rarefactions or both shocks.

Now we find the pressure p^* explicitly. It is easy to see that the functions f_l and f_r in eq. (27) are identical and that $f_l + u_l - V_p = 0$.

If $V_p > u_l$ then both W_L and W_R are rarefaction waves and the solution for p^* is

$$p^* = p_1 \left[1 - \frac{(\gamma - 1)(V_p - u_1)}{2(1 - b\rho_1)C_1} \right]^{\frac{2\gamma}{\gamma-1}} \tag{32}$$

If $V_p \leqslant u_1$ then both W_L and W_R are shock waves with

$$p^* = p_1 \left[\frac{2\alpha_1 + (u_1 - V_p)^2 + (u_1 - V_p)\sqrt{4\alpha_1(1 - \beta) + (u_1 - V_p)^2}}{2\alpha_1} \right] \tag{33}$$

with

$$\alpha_1 = \frac{2(1 - b\rho_1)p_1}{(\gamma + 1)\rho_1} \quad , \quad \beta = \frac{\gamma - 1}{\gamma + 1} \tag{34}$$

For the left boundary the analysis is identical and the result is

$$p^* = p_r \left[1 - \frac{(\gamma - 1)(u_r - V_p)}{2(1 - b\rho_r)C_r} \right]^{\frac{2\gamma}{\gamma-1}} \tag{35}$$

if $V_p < u_r$ (2 rarefactions)

and

$$p^* = p_r \left[\frac{2\alpha_r + V_p - u_r)^2 + (V_p - u_r)\sqrt{4\alpha_r(1 - \beta) + (V_p - u_r)^2}}{2\alpha_r} \right] \tag{36}$$

if $V_p > u_r$ (2 shocks), where α_r is given in eq. (34) with ρ_1, p_1 replaced by ρ_r, p_r.

The problem that remains is the determination of the solution inside rarefaction fans.

4.3 Solution inside rarefaction fans

We only consider one case in detail. Suppose the left travelling wave W_L is a rarefaction wave as illustrated in Fig.3. Consider a general point Q $(\hat{x}, \hat{t})$ inside the rarefaction fan bounded by characteristics $\frac{dx}{dt} = u_1 - c_1$ (head) and $\frac{dx}{dt} = u^* - c_1^*$ (tail). A characteristic ray through the origin and Q has slope $\frac{dx}{dt} = u - c$ in the x - t plane, where both u and c are unknowns of the problem. Then

$$u = \frac{\hat{x}}{\hat{t}} + c \tag{37}$$



Figure 3: Point $Q(\hat{x}, \hat{t})$ inside rarefaction fan centered at (0,0).

Use of the left Riemann invariant J_L given by eq. (17) and of eq. (37) gives

$$c[1 + \frac{2}{(\gamma - 1)} (1 - b\rho)] = J_L(U_1) - \frac{\hat{x}}{\hat{t}} \tag{38}$$

Now using definition (9) of sound speed and isentropic relation (18), with ρ_1^* replaced by ρ, at point Q we obtain

$$p = p_1\left(\frac{1 - b\rho_1}{\rho_1}\right)^{\gamma} \left(\frac{\rho}{1 - b\rho}\right)^{\gamma} \tag{39}$$

Further algebraic manipulations give

$$F_L = \rho^{(\gamma-1)}(\gamma + 1 - 2b\rho)^2 - \beta_1(1 - b\rho)^{\gamma+1} = 0 \tag{40}$$

and

$$\frac{\partial F_L}{\partial \rho} = (\gamma + 1)[b\beta_1(1 - b\rho)^{\gamma} + (\gamma + 1 - 2b\rho)(\gamma - 1 - 2b\rho)\rho^{\gamma-2}] \tag{41}$$

where the constant β_1 is given by

$$\beta_l = \frac{\left\{(\gamma - 1)[J_L(U_l) - \frac{\hat{x}}{\hat{t}}]\right\}^2}{\gamma p_l (\frac{1 - b\rho_l}{\rho_l})} \tag{42}$$

Eq. (40) is a non-linear algebraic equation for ρ. We solve this using a combination of the Newton-Raphson and the Secant Methods. Once ρ is found, to a given accuracy, the pressure p follows immediately from eq. (39). The sound speed c is now known from eq. (9) and velocity u follows directly from eq. (37).

For the case of a right rarefaction the analysis is entirely analogous. The equation for ρ inside the fan is

$$F_R = \rho^{(\gamma-1)}(\gamma + 1 - 2b\rho)^2 - \beta_r(1 - b\rho)^{\gamma+1} = 0 \tag{43}$$

where

$$\beta_r = \frac{[\hat{x}/\hat{t} - u_r + \frac{2c_r}{(\gamma - 1)}(1 - b\rho_r)]^2}{\gamma p_r (\frac{1 - b\rho_r}{\rho_r})^{\gamma}} \tag{44}$$

Then p follows from an equation like eq. (39) with ρ_l, p_l replaced by ρ_r, p_r. The sound speed c follows from the definition (9) and u is given by

$$u = \hat{x}/\hat{t} - c \tag{45}$$

The exact solution of the Riemann problem with constant covolume is now known everywhere in the half x - t plane (Fig. 1).

5. The Random Choice Method (RCM) with covolume

In this section we describe the way the exact solution of the Riemann problem can be used locally to obtain (numerically) the global solution of the general initial-boundary value problem for the Euler equations.

Consider the system of equations (2) in a finite domain $0 \leqslant x \leqslant L$ subject to a general initial data at a time t_n, say. If the spatial domain is discretised into M cells of size Δx and the general data is approximated by piece-wise constant functions then the original problem has been replaced by a sequence of local Riemann problems Rp(i, i+1) for i = 1, ... M - 1. In addition, there are two more boundary Riemann problems RP(0,1) and RP(M, M+1). Data for RP(i, i+1) consists of two constant states U_i^n (left) and U_{i+1}^n (right). The discrete problem is illustrated in Fig. 4. Each local Riemann problem has solution as depicted in Fig. 1 and can be solved exactly by the method of section 4. Now the solution is valid locally for a restricted range of space and time, i.e. before wave interaction occurs. For a sufficiently small time increment ΔT the local solutions are unique in their respective domains so that the global solution at time $t_{n+1} = t_n + \Delta T$ is uniquely defined for $0 \leqslant x \leqslant L$. Within cell i (Fig.4), the solution is composed of the exact solutions of RP(i-1, i) and RP(i, i+1). We denote this exact solution by V_i^{n+1}. Note that $V_i^{n+1}(x, t_{n+1})$ depends on $x(x_i < x < x_{i+1})$; it is not constant, in general. In fact, there may be strong discontinuities transversing cell i. In order to advance the numerical solution in time, a procedure to update U_i^n to U_i^{n+1} is required. The Random Choice Method ([4], [6]) takes

$$U_i^{n+1} = V_i^{n+1}(Q_i) \tag{46}$$

where $Q_i = (x_i + \theta_n \Delta x,\ t_n + \Delta T)$ is a point at a "random" position within cell i. Here θ_n is a pseudo-random number in the interval [0,1].

We remark that a more well known version of RCM advances the solution in two steps using a staggered grid [6]. The one-step RCM on a non-staggered grid as presented here is simpler to implement and has a number of advantages over the staggered-grid version. This is most evident when source terms depending on x and t are incorporated; also when using higher-order versions [7], or hybrid schemes [8], or irregular grids [9], the one-step RCM facilitates coding enormously.

Two more aspects of the method require attention, namely, the choice of the time-step size ΔT and the generation of the pseudo-random numbers θ_n. The choice of ΔT is dictated by the requirement that no waves should interact. This is the CFL condition. A popular version [4] for RCM is

$$\Delta T = C_S\ \Delta x / S_{max} \tag{47}$$



Fig. 4: Solution of local Riemann problems RP(i-1,i) and RP(i,i+1) affecting cell i.

where the coefficient C_S is chosen within the interval $(0,\frac{1}{2}]$ and S_{max} is the maximum wave speed present at time t_n, i.e.

$$S_{max} = \max_i \{|u_i^n| + c_i^n\} \tag{48}$$

The CFL condition (47) chooses ΔT in such a way that no wave is allowed to transverse more than half a cell size. This is convenient to implement, but one could do better by monitoring intersection points within each cell and then choosing ΔT appropriately.

Concerning the sequence $\{\theta_n\}$, it has been established [4] that Van der Corput sequences give best results. Truly random numbers are not as adequate. A general Van der Corput sequence [10] $\{\theta_n\}$ depends on two parameters k_1, k_2 with $k_1 > k_2 > 0$, both integer and relatively prime. Then the (k_1, k_2) van der Corput sequence $\{\theta_n\}$ is formally defined as follows

$$\theta_n = \sum_{i=0}^{m} A_i k_1^{-(i+1)} \tag{49}$$

where

$$A_i = k_2 a_i \pmod{k_1} \tag{50}$$

and

$$n = \sum_{i=0}^{m} a_i \, k_1^i \tag{51}$$

Eq. (49) says that the n-th member $\theta_n \varepsilon[0,1]$ of the (k_1, k_2) van der Corput sequence is a summation of m terms involving powers of k_1. The coefficients A_i are defined by eqs. (50) and (51). First, the non-negative integer n is expressed in scale of notation with radix k_1 (base k_1) by eq. (51). e.g. $k_1 = 2$ gives the binary expansion of n.

Table I contains coefficients a_i of eq. (51) for $k_1 = 2$ and $k_1 = 3$ for ten values of n. The next stage is to find the "modified" coefficients A_i from eq. (50), i.e. A_i is the remainder of dividing $k_2 a_i$ by k_1 ($A_i < k_1$). The simplest case is $k_2 = 1$, then $A_i = a_i \ \forall_i$. Table II(a) shows the coefficients A_i for ten values of n when $k_1 = 3$ and $k_2 = 2$. Having found A_i for i = 0, . . . , m, the actual members θ_n of the sequence are computed from eq. (49). Table II(b) shows the first 10 members of two van der Corput sequences.

	$k_1 = 2$					$k_1 = 3$			
n	a_0	a_1	a_2	a_3	m	a_0	a_1	a_2	m
1	1				0	1			0
2	0	1			2	2			1
3	1	1			2	0	1		2
4	0	0	1		3	1	1		2
5	1	0	1		3	2	1		2
6	0	1	1		3	0	2		2
7	1	1	1		3	1	2		2
8	0	0	0	1	4	2	2		2
9	1	0	0	1	4	0	0	1	3
10	0	1	0	1	4	1	0	1	3

Table I: Coefficients a_i and value of m when $k_1 = 2$ and $k_1 = 3$ for n = 1 to 10

n	A_o	A_1	A_2
1	2		
2	1		
3	0	2	
4	2	2	
5	1	2	
6	0	1	
7	2	1	
8	1	1	
9	0	0	2
10	2	0	2

(a)

θ_n for (2,1)	θ_n for (3,2)
0.0	0.1667
-0.25	-0.1667
0.25	-0.2778
-0.375	0.3889
0.125	0.0556
-0.125	-0.3889
0.375	0.2778
-0.4375	-0.0556
0.0625	-0.4259
-0.1875	0.2407

(b)

Table II: (a) Coefficients A_i for sequence (3,2) and
(b) van der Corput numbers (2,1) and (3,2) for n = 1 to 10

The final stage to implement RCM is the sampling procedure. Fig. 4 shows that the updated value U_i^{n+1} depends on sampling the exact solution of the Riemann problems RP(i-1,i) and RP(i,i+1). Note that for each cell i we only solve one Riemann problem, except for i = 1. Given the CFL condition (47) we sample the right half of the solution of RP(i-1,i) if $0 \leqslant \theta_n \leqslant \frac{1}{2}$ or the left half of the solution of RP(i,i+1) if $\frac{1}{2} < \theta_n \leqslant 1$. The sampling procedure itself, irrespective of the value of θ_n, has two main cases to consider, namely (A) the sampling point Q_i lies to the left of the contact discontinuity $\frac{dx}{dt} = u^*$ and (B) Q_i lies to the right of the contact discontinuity. Each case has two possible wave configurations. Figs. 5 and 6 show these configurations for cases (A) and (B) respectively.

Consider case (A), i.e. Q_i is to the left of $\frac{dx}{dt} = u^*$. The flow chart of Fig. 7 shows the detailed sampling procedure. One proceeds to sample the wave pattern of Fig. 5a if the left wave is a shock wave, i.e. $p^* \geqslant p_1$. Otherwise the wave configuration of Fig. 5b is sampled (left rarefaction). For the shock case there are two possible regions, namely behind the shock (region star left) or in front of the shock (left state). For the rarefaction case there are three possible regions. If Q_i lies to the right of the tail of the rarefaction $\frac{dx}{dt} = u^* - c_1^*$, then we assign the solution



(a) (b)

Figure 5: Wave configuration for case A when Q_i is to the left of contact: (a) W_L is shock, (b) W_L is rarefaction.



(a) (b)

Figure 6: Wave configuration for case B where Q_i is to the right of contact: (a) W_R is shock, (b) W_R is rarefaction.



Figure 7: Sampling procedure for case (A), Q_i lies to the left of contact discontinuity $\frac{dx}{dt} = u^*$ (see Fig. 5)

corresponding to the region star left. If Q_i lies to the left of the rarefaction head $\frac{dx}{dt} = u_1 - c_1$ then the data state U_1 is assigned to the solution. Finally, if Q_i lies inside the rarefaction fan the non-linear eq. (40) must be solved to find ρ; the pressure p is found from eq. (39) and the velocity u is found from eq. (37).

Case (B), Q_i lies to the right of the contact discontinuity, is entirely similar to case (A) just described; it is its mirror image (see Fig. 6).

The application of the solution of the Riemann problem with covolume to the Random Choice Method has been described. The resulting numerical technique to solve the one-dimensional unsteady Euler equations with general data and boundary conditions of practical interest can now be applied to a variety of problems in which covolume is important. Note that the present Riemann solver applies directly to the ideal-gas case ($b = 0$). Indeed, if covolume is not needed, then it is more efficient to exclude covolume in all equations.

In Ref. [3], details of the ideal gas algorithm are given, including FORTRAN programs for the Riemann solver and its implementation in the Random Choice Method.

6. Application to shock-tube problems

Shock-tube problems are special cases of a Riemann problem and can therefore be solved exactly by direct application of the present Riemann solver. Also, as gas dynamical problems they can be solved approximately by solving the Euler equations numerically. This is done here by use of RCM which in turn utilises, locally, the exact solution of the Riemann problem.

First, as a partial validation of the method, we solved the shock-tube problem with data as given in Table IIIa. This is the ideal-gas case ($b \equiv 0$) and has a similarity solution. Fig. 8 shows results. They are coincident, as they should be. The second shock-tube problem is defined by data of Table IIIb. This is a case with covolume. Fig. 9 shows a comparison between the ideal case ($b \equiv 0$) and the non-ideal case ($b = 10^{-3} m^3/kg$).

Differences are relatively small. The ideal gas case gives a stronger shock but a weaker contact discontinuity. Also the rarefaction for the ideal case is slightly weaker, but overall variations in ρ, u, p inside the rarefaction fan are small. Variation in internal energy are appreciable. This has implications for ignition criteria.

Fig. 10 shows a comparison between the exact solution and the numerical solution (obtained by RCM) of the covolume shock tube problem.

(a) ideal case		(b) non-ideal case		
$b = 0.0$		$b = 0.001$		(m^3/kg)
$\gamma = 1.4$		$\gamma = 1.3$		
$\rho_l = 1.0$	, $\rho_r = 0.125$	$\rho_l = 100.0$	$\rho_r = 1.0$	(kg/m^3)
$u_l = 0.0$	, $u_r = 0.0$	$u_l = 0.0$	$u_r = 0.0$	(m/s)
$p_l = 1.0$	, $p_r = 0.1$	$p_o = 100.0$	$p_r = 0.1$	(MPa)
$x_o = 0.4$		$x_o = 0.4$		

Table III: Data for two shock-tube problems.

7. Conclusions

An efficient method for solving exactly the Riemann problem with constant covolume has been presented. The Riemann solver can be directly applied to shock-tube problems. The corresponding ideal-gas version of the Riemann solver is very fast by current standards, Ref. [3].

The solution has been applied to the Random Choice Method to solve numerically the general initial boundary value problem for the unsteady one-dimensional Euler equations with the constant covolume equation of state.



SOD'S SHOCK-TUBE PROBLEM.
FIGURE 8 : PRESENT EXACT SOLUTION (SYMBOL) AND SIMILARITY SOLUTION (FULL LINE).



SHOCK-TUBE PROBLEM WITH COVOLUME B=0.001
FIGURE 9 : EXACT SOLUTIONS FOR COVOLUME CASE (FULL LINE) AND IDEAL CASE B=0 (BROKEN LINE).



SHOCK-TUBE PROBLEM WITH COVOLUME B=0.001
FIGURE 10 : COMPUTED SOLUTION BY THE RANDOM CHOICE METHOD (SYMBOL) AND EXACT SOLUTION (FULL LINE).

References

1. Corner J.
   Theory of the Interior Ballistics of Guns.
   John Wiley and Sons, 1950.

2. Toro, E.F. and Clarke, J.F.
   Applications of the Random Choice Method to computing problems of solid propellant combustion in a closed vessel.
   CoA report NFP 85/16, November 1985, Cranfield Institute of Technology, Cranfield, England.

3. Toro, E.F.
   The Random Choice Method on a non-staggered grid utilising an efficient Riemann solver.
   CoA report No 8708, May 1987, Cranfield Institute of Technology, Cranfield, England.

4. Colella, P.
   Glimm's Method for Gas Dynamics.
   SIAM J. Sci. Stat. Comput, Vol. 3, No. 1, P.76, 1982.

5. Courant R.and Friedricks, K.O.
   Supersonic flow and shock waves.
   Springer-Verlag, 1985.

6. Chorin, A.
   Random Choice Solutions of Hyperbolic Systems.
   J. Comp. Phys., Vol. 22, pp. 517-536, 1976.

7. Toro, E.F.
   A New Numerical Technique for Quasi-linear Hyperbolic Systems of Conservation Laws.
   CoA report No. 86/26, December 1986, Cranfield Institute of Technology, Cranfield, England.

8. Toro, E.F. and Roe, P.L.
A Hybridised Higher-Order Random Choice Method for Quasi-linear Hyperbolic Systems.
Proc. 16th International Symposium on Shock Tubes and Waves. July 26-30 1987, Aachen, W. Germany (to appear).

9. Gottlieb, J.J.
Staggered and non-staggered grids with variable node spacing for the Random Choice Method.
Paper presented at Second International Meeting on Random Choice Methods in Gas Dynamics. College of Aeronautics, Cranfield Institute of Technology, Cranfield, England. July 20-24, 1987.
(submitted to J. Comp. Physics)

10. Hammersley, J.M. and Handscombe, D.C.
Monte Carlo Methods.
Chapman and Hall, 1964.

APPENDIX

A FORTRAN 77 program for the Random Choice Method using the constant covolume Riemann solver is included.

There is a DRIVER (main program) and a set of seven subroutines.

DRIVER Program

There are 3 one-dimensional arrays for density D, velocity U and pressure P. Also there is an array RN that holds the random numbers required for the calculations; if more than 10000 time steps are required then its dimension will have to be changed. The common block CPGAMMA contains various constants involving the ratio of specific heats GAMMA. STATES contains data for Riemann problem $RP(U_\ell,U_r)$. STARSO contains solution u* and p* of Riemann problem as well as sound speeds c_ℓ and c_r. COVOLU contains expressions involving the covolume B and B itself. GAMTOL has GAMMA and the tolerance TOL.

The following data is read in:

TUBLEN	:	length
M	:	It defines spatial discretion (e.g. M=100)
NOTIST	:	Number of time steps (e.g. 200)
NOPROF	:	No. of profiles (times) to be printed out (e.g. 10)
TOL	:	Tolerance for iterative solution procedures.
CFLCOE	:	Coefficient for CFL - condition ($0<CFLCOE<\frac{1}{2}$) in calculating time step size ΔT.
GAMMA	:	Ratio of specific heats γ(e.g. 1.3)
B	:	covolume (e.g. $0.001 m^3/kg$)

Main loop 0001 is for time stepping. Loop 0003 solve M + 1 Riemann problems and updates solution by sampling Riemann problem solutions.

Also, there are the following subroutines:

SUBROUTINES RPCOV - this is our Riemann solver for the constant covolume equation of state.

SUBROUTINE SAMCOV - it samples solution of Riemann problem $RP(U_\ell,U_r)$.

SUBROUTINE RARFAN - computes quantities inside rarefaction fans. It solves iteratively for density ρ first.

SUBROUTINE VDCK12 - this generates random number sequences (k_1,k_2) starting at NRNØ.

SUBROUTINE ICDATA - gives initial condition and calculates various constants to be used throughout the computations.

SUBROUTINE CFLCON - calculates ΔT according to CFL condition.

SUBROUTINE OUTPUT - it prints out ρ,u,p and e for specified times.

```
C-----------------------------------------------------------------------
      PROGRAM DRIVER
      PARAMETER (MD=1000)
      DIMENSION D(0:MD+1),U(0:MD+1),P(0:MD+1),RN(10000),TV(10)
      COMMON/CPGAMMA/GP1,GM1,HGM1,DGAM,G1,G2,G3,G4,G5,G6
      COMMON/STATES/DL,UL,PL,DR,UR,PR
      COMMON/STARSO/US,PS,CL,CR
      COMMON/COVOLU/COVL,COVR,B
      COMMON/GAMTOL/GAMMA,TOL
      DATA NC,TIME,POINTER,TOLTIME/0,0.0,0.5,1.0E-06/
      DATA (TV(KT),KT=1,2)/0.0002,0.0004/
      READ(99,*)TUBLEN,M,NOTIST,NOPROF,TOL,CFLCOE,GAMMA,B
      CALL ICDATA(M,TUBLEN,DX,GAMMA,D,U,P)
      CALL VDCK12(RN,NOTIST)
      KT=1
C     COMMENCE TIME STEPPING
      DO 0001 N=1,NOTIST
C     REFLECTING BOUNDARY CONDITIONS APPLIED
      D(0)  =D(1)
      U(0)  =U(1)
      P(0)  =P(1)
      D(M+1)=D(M)
      U(M+1)=-U(M)
      P(M+1)=P(M)
      CALL CFLCON(B,GAMMA,M,D,U,P,DX,DTMIN)
      DT=CFLCOE*DTMIN
      TITEST=(TIME+DT)
      IF(TITEST.GT.TV(KT))THEN
         DT=TV(KT)-TIME
      ENDIF
      TIME=TIME+DT
      RAND=RN(N)
      DTDX=DT/DX
      DXDTL=RAND/DTDX
      DXDTR=(RAND-1.0)/DTDX
C     UPDATE SOLUTION TO NEXT TIME LEVEL
         DO 0003 I=1,M
            IF(I.EQ.1)THEN
C              SOLVE RIEMANN PROBLEM AT THE LEFT BOUNDARY
               DL=D(I-1)
               UL=U(I-1)
               PL=P(I-1)
               DR=D(I)
               UR=U(I)
               PR=P(I)
               CALL RPCOV
            ENDIF
            IF(RAND.LE.POINTER)THEN
              CALL SAMCOV(D1,U1,P1,DXDTL)
            ENDIF
C     SOLVE RIEMANN PROBLEM RP(I,I+1)
            DL=D(I)
            UL=U(I)
            PL=P(I)
            DR=D(I+1)
```

```
             UR=U(I+1)
             PR=P(I+1)
             CALL RPCOV
             IF(RAND.GT.POINTER)THEN
                CALL SAMCOV(D1,U1,P1,DXDTR)
             ENDIF
             D(I)=D1
             U(I)=U1
             P(I)=P1
 0003     CONTINUE
C      UPDATING COMPLETED
       TDIF=ABS(TIME-TV(KT))
       IF(TDIF.LE.TOLTIME)THEN
          NC=NC+1
          CALL OUTPUT(TIME,M,NC,NOPROF,GM1,D,U,P,B)
          IF(NC.EQ.0)THEN
             WRITE(6,*)'JOB FINISHED OK'
             STOP
          ENDIF
          KT=KT+1
       ENDIF
 0001  CONTINUE
C      TIME STEPPING COMPLETED
       END
C-----------------------------------------------------------------
       SUBROUTINE RPCOV
       COMMON/STATES/DL,UL,PL,DR,UR,PR
       COMMON/STARSO/US,PS,CL,CR
       COMMON/GAMTOL/GAMMA,TOL
       COMMON/COVOLU/COVL,COVR,B
       COMMON/CPGAMMA/GP1,GM1,HGM1,DGAM,G1,G2,G3,G4,G5,G6
C      SOLVES RIEMANN PROBLEM WITH CONSTANT COVOLUME B
       COVL=1.0-B*DL
       COVR=1.0-B*DR
       CL   =SQRT(GAMMA*PL/(COVL*DL))
       CR   =SQRT(GAMMA*PR/(COVR*DR))
       DELU=UL-UR
C      GUESSED VALUE FOR PS IS PROVIDED
       CLPLG=CL/PL**G1
       CRPRG=CR/PR**G1
       ABOVE=CL*COVL+CR*COVR+HGM1*DELU
       BELOW=CLPLG*COVL+CRPRG*COVR
       PS   =(ABOVE/BELOW)**G3
       PS0  =PS
C      START ITERATION
       DO 0001 IT=1,50
C         LEFT WAVE
          IF(PL.LT.PS)THEN
             S1=SQRT(G5*COVL/DL)
             S2=G6*PL
             S2PS=S2+PS
             DELPLPS=PL-PS
             SQS2PS=1.0/SQRT(S2PS)
             FLEFVAL=S1*DELPLPS*SQS2PS
             FLEFDER=-S1*SQS2PS*(1.0+0.5*DELPLPS/S2PS)
```

```
            ELSE
               FLEFVAL=G4*COVL*(CL-CLPLG*PS**G1)
               FLEFDER=-DGAM*COVL*CLPLG*PS**(-G2)
            ENDIF
C        RIGHT WAVE
            IF(PR.LT.PS)THEN
               S1=SQRT(G5*COVR/DR)
               S2=G6*PR
               S2PS=S2+PS
               DELPRPS=PR-PS
               SQS2PS=1.0/SQRT(S2PS)
               FRIGVAL=S1*DELPRPS*SQS2PS
               FRIGDER=-S1*SQS2PS*(1.0+0.5*DELPRPS/S2PS)
            ELSE
               FRIGVAL=G4*COVR*(CR-CRPRG*PS**G1)
               FRIGDER=-DGAM*COVR*CRPRG*PS**(-G2)
            ENDIF
            FUNVAL=FLEFVAL+FRIGVAL+DELU
            FUNDER=FLEFDER+FRIGDER
            PS     =PS-FUNVAL/FUNDER
            IF(IT.GT.5)THEN
C              SECANT METHOD
               ABOVE=PS0*FUNVAL-PS*FUNVAL0
               BELOW=FUNVAL-FUNVAL0
               PS=ABOVE/BELOW
            ELSE
C              NEWTON RAPHSON METHOD
            ENDIF
            US=0.5*(FLEFVAL-FRIGVAL+UL+UR)
            TESTPS  =ABS((PS-PS0)/PS)
            IF(TESTPS.LE.TOL)GOTO 0002
            IF(PS.LT.TOL)PS=TOL
            PS0=PS
            FUNVAL0=FUNVAL
 0001    CONTINUE
         WRITE(6,0003)IT
         STOP
 0003    FORMAT(' DIVERGENCE IN PSTAR STEP, ITERATION  NO. =',I4)
 0002    CONTINUE
         RETURN
         END
C----------------------------------------------------------------------
         SUBROUTINE SAMCOV(D,U,P,DXDT)
         COMMON/STATES/DL,UL,PL,DR,UR,PR
         COMMON/STARSO/US,PS,CL,CR
         COMMON/COVOLU/COVL,COVR,B
         COMMON/GAMTOL/GAMMA,TOL
         COMMON/CPGAMMA/GP1,GM1,HGM1,DGAM,G1,G2,G3,G4,G5,G6
         IF(DXDT.GE.US)THEN
C        SAMPLING POINT LIES TO THE RIGHT OF SLIP LINE
            IF(PS.LE.PR)THEN
C           RIGHT WAVE IS A RAREFACTION WAVE
               IF(DXDT.LT.(UR+CR))THEN
                  AISEN=(DR/COVR)*(PS/PR)**DGAM
                  D3    =AISEN/(1.0+B*AISEN)
```

```
                COV3 =1.0-B*D3
                C3   =SQRT(GAMMA*PS/(D3*COV3))
                IF(DXDT.LT.(US+C3))THEN
C               LEFT OF RIGHT RAREFACTION
                   D=D3
                   U=US
                   P=PS
                ELSE
C               INSIDE RIGHT RAREFACTION
C                  GUESS VALUE FOR D, MEAN VALUE
                   D=0.5*(DR+D3)
                   RARCON=DXDT-UR
                   CALL RARFAN(DXDT,RARCON,D,C4,P,DR,PR,CR,COVR)
                   U=DXDT-C4
                ENDIF
             ELSE
C            RIGHT OF RIGHT RAREFACTION
                D=DR
                U=UR
                P=PR
             ENDIF
          ELSE
C         RIGHT WAVE IS A SHOCK WAVE
             CONS=0.5*GP1*DR*PR/COVR
             PRERAT=PS/PR
             RMR=SQRT(CONS*(PRERAT+GM1/GP1))
             URS=UR+RMR/DR
             IF(DXDT.GE.URS)THEN
C            RIGHT OF RIGHT SHOCK
                D=DR
                U=UR
                P=PR
             ELSE
C            BEHIND RIGHT SHOCK
                ABOVE=GP1*PRERAT+GM1
                TWIBDR=2.0*B*DR
                BELOW=(GM1+TWIBDR)*PRERAT+GP1-TWIBDR
                D=DR*ABOVE/BELOW
                U=US
                P=PS
             ENDIF
          ENDIF
       ELSE
C      SAMPLING POINT LIES TO THE LEFT OF SLIP LINE
          IF(PS.LE.PL)THEN
C            LEFT WAVE IS A RAREFACTION
             AISEN=(DL/COVL)*(PS/PL)**DGAM
             D3   =AISEN/(1.0+B*AISEN)
             COV3 =1.0-B*D3
             C3   =SQRT(GAMMA*PS/(D3*COV3))
             IF(DXDT.LT.(US-C3))THEN
                IF(DXDT.LT.(UL-CL))THEN
C               LEFT OF LEFT RAREFACTION
                   D=DL
                   U=UL
```

```
                  P=PL
               ELSE
C              INSIDE LEFT RAREFACTION
C                 GUESS VALUE FOR D, MEAN VALUE
                  D=0.5*(DL+D3)
                  RARCON=-(DXDT-UL)
                  CALL RARFAN(DXDT,RARCON,D,C4,P,DL,PL,CL,COVL)
                  U=DXDT+C4
               ENDIF
            ELSE
C        RIGHT OF LEFT RAREFACTION
               D=D3
               U=US
               P=PS
            ENDIF
         ELSE
C        LEFT WAVE IS A SHOCK WAVE
            CONS=0.5*GP1*DL*PL/COVL
            PRERAT=PS/PL
            RML=SQRT(CONS*(PRERAT+GM1/GP1))
            ULS=UL-RML/DL
            IF(DXDT.GE.ULS)THEN
C           BEHIND LEFT SHOCK
               ABOVE=GP1*PRERAT+GM1
               TWIBDL=2.0*B*DL
               BELOW=(GM1+TWIBDL)*PRERAT+GP1-TWIBDL
               D=DL*ABOVE/BELOW
               U=US
               P=PS
            ELSE
C           LEFT OF LEFT SHOCK
               D=DL
               U=UL
               P=PL
            ENDIF
         ENDIF
      ENDIF
      RETURN
      END
C-----------------------------------------------------------------
      SUBROUTINE RARFAN(DXDT,RARCON,DF,C4,P,DK,PK,CK,COVK)
      COMMON/COVOLU/COVL,COVR,B
      COMMON/GAMTOL/GAMMA,TOL
      COMMON/CPGAMMA/GP1,GM1,HGM1,DGAM,G1,G2,G3,G4,G5,G6
      Z1=RARCON+2.0*CK*COVK/GM1
      Z2=PK*(COVK/DK)**GAMMA
      ZZ=(Z1*GM1)**2/(GAMMA*Z2)
      DF0=DF
      DO 0001 I=1,100
         COVF=1.0-B*DF
         F1  =GP1-2.0*B*DF
         F2  =COVF**GAMMA
         F3  =F1-2.0
         F4  =DF**GM1
         FVAL=F1*F1*F4-ZZ*F2*COVF
```

```
C           NEWTON-RAPHSON ITERATION
            FDER=GP1*(B*ZZ*F2+F1*F3*F4/DF)
            DF  =DF-FVAL/FDER
            IF(I.GT.5)THEN
C              SECANT METHOD
               ABOVE=DF0*FVAL-DF*FVAL0
               BELOW=FVAL-FVAL0
               DF   =ABOVE/BELOW
            ENDIF
            DETED=ABS((DF-DF0)/DF)
            IF(DETED.LE.TOL)GOTO 0002
            IF(DF.LT.TOL)DF=TOL
            DF0  =DF
            FVAL0=FVAL
 0001    CONTINUE
         WRITE(6,0004)I
 0004    FORMAT(5X,'DIRVERGENCE INSIDE FAN, NO. OF ITER.=',I5)
         STOP
C        COMPUTE OTHER UNKNOWNS
 0002    COV4=1.0-B*DF
         P   =Z2*(DF/COV4)**GAMMA
         C4  =SQRT(GAMMA*P/(DF*COV4))
 0003    CONTINUE
         RETURN
         END
C-----------------------------------------------------------------
         SUBROUTINE VDCK12(RN,NOTIST)
         PARAMETER (N1=1000,N2=10000)
         DIMENSION NA(N1),JA(N1),RN(N2)
         DATA K1,K2,NRN0/2,1,100/
         DO 0001 NRN=NRN0,NOTIST+NRN0
            IS=0
            MM=NRN
            DO 0002 I=1,100
               IF(MM.EQ.0)GOTO 8888
               IS=IS+1
               NA(I)=MOD(MM,K1)
               MM=MM/K1
               KL=K2*NA(I)
               JA(I)=MOD(KL,K1)
 0002       CONTINUE
 8888       RANNUM=0.0
            DO 0004 K=1,IS
               RANNUM=RANNUM+REAL(JA(K))/(K1**K)
 0004       CONTINUE
            NT=NRN-NRN0+1
            RN(NT)=RANNUM
 0001    CONTINUE
         RETURN
         END
C-----------------------------------------------------------------
         SUBROUTINE ICDATA(M,TUBLEN,DX,GAMMA,D,U,P)
         PARAMETER (MD=1000)
         DIMENSION D(0:MD+1),U(0:MD+1),P(0:MD+1)
         COMMON/CPGAMMA/GP1,GM1,HGM1,DGAM,G1,G2,G3,G4,G5,G6
```

```
      DATA DL0,UL0,PL0/100.0,0.0,100.0E+06/
      DATA DR0,UR0,PR0/1.0,0.0,0.1E+06/
      DATA X0/0.4/
      GP1=GAMMA+1.0
      GM1=GAMMA-1.0
      HGM1=0.5*GM1
      HGP1=0.5*GP1
      DGAM=1.0/GAMMA
      G1=HGM1/GAMMA
      G2=HGP1/GAMMA
      G3=1.0/G1
      G4=1.0/HGM1
      G5=2.0/GP1
      G6=GM1/GP1
      DX=TUBLEN/REAL(M)
      DO 1000 I=1,M
         XP=(REAL(I)-0.5)*DX
         IF(XP.LE.X0)THEN
            D(I)=DL0
            U(I)=UL0
            P(I)=PL0
         ELSE
            D(I)=DR0
            U(I)=UR0
            P(I)=PR0
         ENDIF
 1000 CONTINUE
      RETURN
      END
C-----------------------------------------------------------------
      SUBROUTINE CFLCON(B,GAMMA,M,D,U,P,DX,DTMIN)
      PARAMETER (MD=1000)
      DIMENSION D(0:MD+1),U(0:MD+1),P(0:MD+1)
      SMAX=0.0
      DO 0001 I=1,M
         DENS=D(I)
         COV=1.0-B*DENS
         A=SQRT(GAMMA*P(I)/(COV*DENS))
         SMUA=ABS(U(I))+A
         IF(SMUA.GT.SMAX)SMAX=SMUA
 0001 CONTINUE
      DTMIN=DX/SMAX
      RETURN
      END
C-----------------------------------------------------------------
      SUBROUTINE OUTPUT(TIME,M,NC,NOPROF,GM1,D,U,P,B)
      PARAMETER (MD=1000)
      DIMENSION D(0:MD+1),U(0:MD+1),P(0:MD+1)
      DIMENSION TM(20),R1(4,20,MD)
      DATA RMPA/1.0E+06/
      TM(NC)=TIME
         GMCONST=GM1*RMPA
         DO 0001 I=1,M
             R1(1,NC,I)=D(I)
             R1(2,NC,I)=U(I)
```

```
            R1(3,NC,I)=P(I)/RMPA
            COV=1.0-B*D(I)
            R1(4,NC,I)=(COV*P(I))/(D(I)*GMCONST)
 0001     CONTINUE
          IF(NC.EQ.NOPROF)THEN
             WRITE(1,0004)(TM(J),J=1,NOPROF)
             WRITE(2,0004)(TM(J),J=1,NOPROF)
             WRITE(3,0004)(TM(J),J=1,NOPROF)
             WRITE(4,0004)(TM(J),J=1,NOPROF)
             DO 0002 I=1,M
                 WRITE(1,0003)I,(R1(1,J,I),J=1,NOPROF)
                 WRITE(2,0003)I,(R1(2,J,I),J=1,NOPROF)
                 WRITE(3,0003)I,(R1(3,J,I),J=1,NOPROF)
                 WRITE(4,0003)I,(R1(4,J,I),J=1,NOPROF)
 0002        CONTINUE
             NC=0
          ENDIF
 0003   FORMAT(I4,1X,10(F10.4,1X))
 0004   FORMAT(5X,10(F7.4,4X))
        RETURN
        END
C-------------------------------------------------------------------
1.0       100      7000       1     1.0E-04     0.4       1.3  0.001

            VARIABLE NAMES FOR TEST PROBLEM WITH COVOLUME

TUBLEN    M       NOTIST  NOPROF    TOL       CFLCOE     GAMMA  B
C-------------------------------------------------------------------
```